



ESKİŞEHİR OSMANGAZI ÜNİVERSİTESİ

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: ogu.ahmet.topcu@gmail.com

Web: <http://mmf2.ogu.edu.tr/atopcu>

Bilgisayar Destekli Nümerik Analiz

Ders notları 2014

Ahmet TOPÇU

```
DEFINT I-N
DEFDBL A-H, O-Z
SUB PowerTers (n, a(), aLandaMin, x(), iHata)
' Mutlak değerce en küçük özdeğer ve özvektörünün hesabı
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi Eskişehir, 1997
' n: Matrisin boyutu
' A(n,n): en küçük özdeğeri hesaplanacak matris
' aLandaMin: hesaplanan özdeğer
' x(n): özvektör
' iHata=-2: Yakınsamadı
' iHata=-1: matris tekil
' iHata=0: Yakınsadı
' Çağırılan programda A depolanmış ve x boyutlandırılmış olmalıdır
' Çağırılan alt programlar: GaussLU, AsagiYukariHesap
'
' DIM iSatic(n), b(n), y(n): ' çalışma vektörleri
'
' Maxit = 200: ' maksimum iterasyon sayısı
' Eps = 1E-14: ' istenilen hassasiyet
' iHata = 0
'
' Start vektör
' FOR i = 1 TO n
'   x(i) = 1
' NEXT i
' A=LU çarpanlarına ayır
' CALL GaussLU(n, a(), iSatic(), iHata)
' IF iHata = -1 THEN EXIT SUB
'
' iIterate = 1
' iterasyona başla
' DO
'   FOR i = 1 TO n
'     b(i) = x(i)
'   NEXT i
'
'   LUy=b den y i hesapla
'   CALL AsagiYukariHesap(n, a(), iSatic(), b(), y())
'   iMax = 1
'   FOR i = 2 TO n
'     IF ABS(x(i)) > ABS(x(iMax)) THEN iMax = i
'   NEXT i
'   x = y
'   iIterate = iIterate + 1
'   IF iIterate > Maxit THEN EXIT DO
'   IF ABS(aLandaMin - x(iMax)) < Eps THEN EXIT DO
'   aLandaMin = x(iMax)
' END DO
'
' RETURN
```

29

PROGRAMLAR: Standart özdeğer ve özvektör hesabı

- En küçük özdeğer ve vektörü - PowerTers

29. PROGRAMLAR: Standart özdeğer ve özvektör hesabı¹ - PowerTers

Bazı mühendislik problemlerinde **mutlak değerce en küçük özdeğerin** ve buna ait normalleştirilmiş özvektörün hesaplanması gerekir (Örnek: bir yapının birinci periyodunun ve titreşim modunun hesabı).

PowerTers alt programı

$$\underline{A}\underline{x} = \lambda\underline{x}$$

standart özdeğer probleminin **mutlak değerce en küçük özdeğerini** ve buna ait normalleştirilmiş özvektörünü Power ters (diğer adı: Mises invers) iterasyon yöntemi ile hesaplar. $\underline{A}_{n \times n}$ simetrik olmayabilir.

\underline{A} simetrik ve pozitif tanımlı ise Power ters iterasyon yöntemi daima yakınsar. Ancak, aşağıdaki durumlarda nümerik zorluk yaratabilir ve yakınsamayabilir:

- \underline{A} simetrik değilse
- Tüm özdeğerler sanal ise
- Tekrarlanan özdeğerler varsa: $|\lambda_i| = |\lambda_j|$
- Özdeğerler birbirine çok yakın ise: $|\lambda_i| \approx |\lambda_j|$

PowerTers alt programı $\underline{A}(n,n)$ matrisini çağıran programda depolanmış ve $\underline{x}(n)$ vektörünün tanımlanmış olduğunu varsayar.

Örnek:

$$\underline{A} = \begin{bmatrix} 10 & 2 & 6 & 1 & 0 \\ 2 & 0 & 4 & 1 & 2 \\ 7 & -4 & 10 & 5 & 1 \\ -3 & 1 & -5 & 5 & 3 \\ 4 & 3 & 1 & 3 & 5 \end{bmatrix}, \text{ en küçük özdeğer ve buna ait özvektör ?}$$


PowerTers alt programının sonucu

```

C:\Basic\QBASIC.EXE
En küçük özdeğer ve özvektörü(PowerTers):
aLamdaMin= 1.16993472375839
-.885859545553477 1 .86612572551689 .625443259717596 -.574146173459904
İterasyon sayısı= 30
  
```

Mutlak değerce en küçük özdeğer: $\lambda = 1.169935$. λ ya ait özvektör: $\underline{x} = \begin{bmatrix} -0.8859 \\ 1 \\ 0.8661 \\ 0.6254 \\ -0.5742 \end{bmatrix}$

¹ Teori ve örnekler için bak: bölüm 25



```

SUB PowerTers (n, a(), aLamdaMin, x(), iHata)
'-----
' Mutlak değerce en küçük özdeğer ve özvektörünün hesabı
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi Eskişehir, 1997
' n: Matrisin boyutu
' A(n,n): en küçük özdeğeri hesaplanacak matris
' aLamdamin: hesaplanan özdeğer
' x(n): özvektör
' iHata=-2: Yakınsamadı
' iHata=-1: matris tekil
' iHata=0: Yakınsadı
' Çağırılan programda A depolanmış ve x boyutlandırılmış olmalıdır
' Çağırılan alt programlar: GaussLU, AsagiYukariHesap
'-----
  DIM iSatir(n), b(n), y(n): ' çalışma vektörleri

  Maxit = 200: ' maksimum iterasyon sayısı
  Eps = 1E-14: ' istenilen hassasiyet
  iHata = 0

' Start vektör
  FOR i = 1 TO n
    x(i) = 1
  NEXT i

' A=LU çarpanlarına ayır
  CALL GaussLU(n, a(), iSatir(), iHata)
  IF iHata = -1 THEN EXIT SUB

  iTeRate = 1
' iterasyona başla
  DO
    FOR i = 1 TO n
      b(i) = x(i)
    NEXT i

' LUy=b den y i hesapla
    CALL AsagiYukariHesap(n, a(), iSatir(), b(), y())
    iMax = 1
    FOR i = 2 TO n
      IF ABS(y(i)) > ABS(y(iMax)) THEN iMax = i
    NEXT i
    yMax = 1 / y(iMax)

' özvektörü normalleştir, erişilen hassasiyeti belirle
    EpsHesap = 0
    FOR i = 1 TO n
      y(i) = y(i) * yMax
      T = x(i) - y(i)
      IF ABS(T) > EpsHesap THEN EpsHesap = ABS(T)
      x(i) = y(i)
    NEXT i

' özdeğer
    iTeRate = iTeRate + 1
    LOOP UNTIL EpsHesap < Eps OR iTeRate > Maxit
    iHata = iTeRate
    aLamdaMin = yMax
    IF iTeRate > Maxit THEN iHata = -2

END SUB ' PowerTers

```

PowerTers alt programı

```

SUB AsagiYukariHesap (n, a(), iSatir(), b(), y())
'-----
' GaussLU ile A=LU çarpanlarına ayrılmış varsayılar
' LUy=b denlem sistemi çözülür
' Yukarıdan aşağı hesap
'-----
  FOR i = 1 TO n - 1
    isi = iSatir(i)
    FOR j = i + 1 TO n
      isj = iSatir(j)
      b(isj) = b(isj) - a(isj, i) * b(isi)
    NEXT j
  NEXT i

' Aşağıdan yukarı hesap
  isi = iSatir(n)
  y(n) = b(isi) / a(isi, n)
  FOR i = 1 TO n - 1
    j = n - i
    isj = iSatir(j)
    y(j) = b(isj)
    FOR k = j + 1 TO n
      y(j) = y(j) - a(isj, k) * y(k)
    NEXT k
    y(j) = y(j) / a(isj, j)
  NEXT i

END SUB ' AsagiYukariHesap

```

AsagiYukari alt programı

GaussLU alt programı

```

SUB GaussLU (n, a(), iSatir(), iHata)
'-----
' A matrisi üçgen çarpanlarına ayrılır: A=LU
' iSatir: satır değiştirme vektörü
' iHata=-1 ile dönerse A tekildir
'-----
' Machepe
  Eps = 1
  DO
    Eps = Eps / 2
    s = 1 + Eps
    LOOP UNTIL s <= 1
    Eps = 2 * Eps
    zero = Eps

    FOR i = 1 TO n
      iSatir(i) = i
    NEXT i
    FOR i = 1 TO n - 1

' Pivot satırını belirle
      iV = i
      FOR j = i + 1 TO n
        isi = iSatir(iV)
        isj = iSatir(j)
        IF (ABS(a(isj, i)) > ABS(a(isi, i))) THEN iV = j
      NEXT j

' Matris tekil mi?
      IF (ABS(a(iSatir(iV), i)) <= zero) THEN
        iHata = -1: ' Matris tekil
        EXIT SUB
      END IF

' satır değişimi vektörü
      IF (iSatir(i) <> iSatir(iV)) THEN
        isi = iSatir(i)
        iSatir(i) = iSatir(iV)
        iSatir(iV) = isi
      END IF

' indirge
      isi = iSatir(i)
      FOR j = i + 1 TO n
        isj = iSatir(j)
        a(isj, i) = a(isj, i) / a(isi, i)
        FOR k = i + 1 TO n
          a(isj, k) = a(isj, k) - a(isj, i) * a(isi, k)
        NEXT k
      NEXT j
    NEXT i
  4
  IF ABS(a(iSatir(n), n)) <= zero THEN iHata = -1

END SUB ' GaussLU

```